PrSAT Demo

Branden Fitelson
October 8, 2014

PrSAT is a *Mathematica* package I have written, which supports algebraic reasoning about the probability calculus. It is freely available from:

http://fitelson.org/PrSAT/

There is also a journal paper that goes along with **PrSAT**, which explains the basic ideas behind its operation. That paper can be downloaded from:

http://fitelson.org/pm.pdf

In this demo, I will go through some homework exercises (HW #2).

HW #2 Problems

First, we load in the **PrSAT** package.

In[1]:=
```
<< PrSAT`
```

Problem 2.5

Mike specifies a probability distribution over the states involving three atomic sentences: **P,Q,R**. We can have **PrSAT** compute this model, by specifying all of these state-probabilities. **PrSAT** takes a set of probabilistic constraints as input and outputs a probability model (iff there is one) satisfying those constraints. Here's what input specifying the distribution from Problem 2.5 looks like. Note: *Mathematica* uses the symbol ∧ (rather than &) for conjunction; ¬ (rather than ~) for negation; and Pr(•), rather than cr(•), for the probability function. Also, *Mathematica* uses brackets rather than parentheses for functions (*e.g.*, Pr[•]); and it uses the symbol "==" for equational constraints (rather than "=").

In[2]:=
```
Model25 = PrSAT[{
    Pr[P ⋀ Q ⋀ R] == 0.1,
    Pr[P ⋀ Q ⋀ ¬ R] == 0.2,
    Pr[P ⋀ ¬ Q ⋀ R] == 0,
    Pr[P ⋀ ¬ Q ⋀ ¬ R] == 0.3,
    Pr[¬ P ⋀ Q ⋀ R] == 0.1,
    Pr[¬ P ⋀ Q ⋀ ¬ R] == 0.2,
    Pr[¬ P ⋀ ¬ Q ⋀ R] == 0,
    Pr[¬ P ⋀ ¬ Q ⋀ ¬ R] == 0.1
  }]
```

Out[2]=
$$\{\{P \to \{a_2, a_5, a_6, a_8\}, Q \to \{a_3, a_5, a_7, a_8\}, R \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\}, \{a_1 \to 0.1, a_2 \to 0.3, a_3 \to 0.2, a_4 \to 0., a_5 \to 0.2, a_6 \to 0., a_7 \to 0.1, a_8 \to 0.1\}\}$$

We can use the **TruthTable** function to visualize the stochastic truth-table for the above probability model.

In[3]:= `TruthTable[Model25]`

Out[3]=

P	Q	R	var	Pr
T	T	T	a_8	0.1
T	T	F	a_5	0.2
T	F	T	a_6	0.
T	F	F	a_2	0.3
F	T	T	a_7	0.1
F	T	F	a_3	0.2
F	F	T	a_4	0.
F	F	F	a_1	0.1

Now, we can use the **`EvaluateProbability`** function to calculate probabilities of any propositions, under this probability distribution (which we have named **`Model25`**).

(a) cr(P ≡ Q)

Mathematica uses the symbol ⟺ for the biconditional. Also, *Mathematica* uses the symbol ⇒ for the material conditional. We can define our own operators here, so as to maintain our familiar notations:

In[4]:=
```
p_ ⊃ q_ := p ⇒ q;
p_ ≡ q_ := p ⟺ q;
```

Now, we can use **`EvaluateProbability`** to calculate the value of cr(P ≡ Q), according to **`Model25`**.

In[6]:= `EvaluateProbability[Pr[P ≡ Q], Model25]`

Out[6]= 0.4

(b) cr(R ⊃ Q)

In[7]:= `EvaluateProbability[Pr[R ⊃ Q], Model25]`

Out[7]= 1.

(c) cr(P & R) – cr(~P & R)

Recall, *Mathematica* uses ∧ for conjunction and ¬ for negation.

In[8]:= `EvaluateProbability[Pr[P ⋀ R] - Pr[¬ P ⋀ R], Model25]`

Out[8]= 0.

(d) cr(P & Q & R) / cr(R)

In[9]:= `EvaluateProbability[Pr[P ⋀ Q ⋀ R] / Pr[R], Model25]`

Out[9]= 0.5

Problem 2.6

Can a probabilistic credence distribution assign cr(P) = 0.5, cr(Q) = 0.5, and cr(~P & ~Q) = 0.8. **No**.

In[10]:=
```
PrSAT[{
   Pr[P] == 1/2,
   Pr[Q] == 1/2,
   Pr[¬ P ⋀ ¬ Q] == 8/10
 }]
```

PrSAT :: srchfail : Search phase failed ; attempting FindInstance

Out[10]= { }

Undocumented feature: **PrReduce** — finds the general conditions under which a set of probabilistic conditions is satisfied.

In[11]:=
```
PrReduce[{
   Pr[P] == 1/2,
   Pr[Q] > Pr[P ⋀ Q],
   Pr[¬ P ⋀ ¬ Q] < 8/10
 }]
```

Out[11]= $a_3 > 0 \;\&\&\; a_2 + a_4 == \frac{1}{2}$

Problem 2.7

Can a probability distribution satisfy all four of the following constraints?

1. Pr(A ⊃ (B ≡ *C*)) = 1.
2. Pr(B) = Pr(~B).
3. Pr(*C*) = 2*Pr(*C* & A).
4. Pr(B & *C* & ~A) = 1/5.

Yes.

In[12]:=
```
Model27 = PrSAT[{
    Pr[A ⊃ (B ≡ C)] == 1,
    Pr[B] == Pr[¬ B],
    Pr[C] == 2 * Pr[C ⋀ A],
    Pr[B ⋀ C ⋀ ¬ A] == 1/5
  }, BypassSearch → True]
```

Out[12]= $\{\{A \to \{a_2, a_5, a_6, a_8\}, B \to \{a_3, a_5, a_7, a_8\}, C \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\}, \{a_1 \to 0, a_2 \to \frac{2}{5}, a_3 \to 0, a_4 \to \frac{1}{10}, a_5 \to 0, a_6 \to 0, a_7 \to \frac{1}{5}, a_8 \to \frac{3}{10}\}\}$

In[13]:=
```
TruthTable[Model27]
```

Out[13]=

A	B	C	var	Pr
T	T	T	a_8	$\frac{3}{10}$
T	T	F	a_5	0
T	F	T	a_6	0
T	F	F	a_2	$\frac{2}{5}$
F	T	T	a_7	$\frac{1}{5}$
F	T	F	a_3	0
F	F	T	a_4	$\frac{1}{10}$
F	F	F	a_1	0

We can use **EvaluateProbability** to check that our model above has the four desired properties.

In[14]:=
```
EvaluateProbability[{
   Pr[A ⊃ (B ≡ C)] == 1,
   Pr[B] == Pr[¬ B],
   Pr[C] == 2 * Pr[C ⋀ A],
   Pr[B ⋀ C ⋀ ¬ A] == 1/5
  }, Model27]
```

Out[14]= {True, True, True, True}

Two Problems from Chapter 3 (not on HW #3)

Problem 3.11

This problem involves showing that the following four probabilistic constraints:

1. Pr(A | *C*) > Pr(A | ¬*C*)
2. Pr(B | *C*) > Pr(B | ¬*C*)
3. Pr(A & B | *C*) = Pr(A | *C*) * Pr(B | *C*)

4. Pr(A & B | ¬*C*) = Pr(A | ¬*C*) * Pr(B | ¬*C*)

jointly imply the following fifth constraint:

5. Pr(A & B) > Pr(A) * Pr(B)

This is equivalent to showing that *there is **no** probability distribution* that satisfies (1)-(4), *but also* satisfies:

6. Pr(A & B) ≤ Pr(A) * Pr(B)

We can verify this claim (of Reichenbach's) using **PrSAT**, which tells us there are no such distributions.

In[15]:=
```
PrSAT[{
   Pr[A | C] > Pr[A | ¬ C],
   Pr[B | C] > Pr[B | ¬ C],
   Pr[A ⋀ B | C] == Pr[A | C] * Pr[B | C],
   Pr[A ⋀ B | ¬ C] == Pr[A | ¬ C] * Pr[B | ¬ C],
   Pr[A ⋀ B] ≤ Pr[A] * Pr[B]
  }]
```

PrSAT :: srchfail : Search phase failed ; attempting FindInstance

Out[15]= { }

Here is a model satisfying all of Reichenbach's conditions:

In[161]:=
```
ModelReich = PrSAT[{
    Pr[A | C] > Pr[A | ¬ C],
    Pr[B | C] > Pr[B | ¬ C],
    Pr[A ⋀ B | C] == Pr[A | C] * Pr[B | C],
    Pr[A ⋀ B | ¬ C] == Pr[A | ¬ C] * Pr[B | ¬ C],
    Pr[A ⋀ B] > Pr[A] * Pr[B]
   }]
```

Out[16]= $\Big\{\{A \to \{a_2, a_5, a_6, a_8\}, B \to \{a_3, a_5, a_7, a_8\}, C \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\}, \left\{a_1 \to \frac{1}{14}, a_2 \to \frac{3}{14}, a_3 \to \frac{1}{14}, a_4 \to 0, a_5 \to \frac{3}{14}, a_6 \to \frac{1}{7}, a_7 \to 0, a_8 \to \frac{2}{7}\right\}\Big\}$

In[17]:= **TruthTable[ModelReich]**

Out[17]=

A	B	*C*	var	Pr
T	T	T	a_8	$\frac{2}{7}$
T	T	F	a_5	$\frac{3}{14}$
T	F	T	a_6	$\frac{1}{7}$
T	F	F	a_2	$\frac{3}{14}$
F	T	T	a_7	0
F	T	F	a_3	$\frac{1}{14}$
F	F	T	a_4	0
F	F	F	a_1	$\frac{1}{14}$

In[18]:=
```
EvaluateProbability[{
   Pr[A | C] > Pr[A | ¬ C],
   Pr[B | C] > Pr[B | ¬ C],
   Pr[A ⋀ B | C] == Pr[A | C] * Pr[B | C],
   Pr[A ⋀ B | ¬ C] == Pr[A | ¬ C] * Pr[B | ¬ C],
   Pr[A ⋀ B] > Pr[A] * Pr[B]
  }, ModelReich]
```

Out[18]= {True, True, True, True, True}

Problem 3.12

(a) find a probability model/distribution according to which:

Pr(*C* | A) = Pr(*C* | B), but
Pr(*C* | A) ≠ Pr(*C* | A v B)

Easily solved with **PrSAT**.

In[19]:=
```
Model312 = PrSAT[{
    Pr[C | A] == Pr[C | B],
    Pr[C | A] ≠ Pr[C | A ⋁ B]
   }]
```

Out[19]= $\{\{A \to \{a_2, a_5, a_6, a_8\}, B \to \{a_3, a_5, a_7, a_8\}, C \to \{a_4, a_6, a_7, a_8\}, \Omega \to \{a_1, a_2, a_3, a_4, a_5, a_6, a_7, a_8\}\}, \{a_1 \to \frac{92\,702\,054\,723}{890\,539\,323\,300}, a_2 \to \frac{1}{25}, a_3 \to \frac{27}{44}, a_4 \to \frac{3}{41}, a_5 \to \frac{4}{61}, a_6 \to \frac{1}{540}, a_7 \to \frac{5}{57}, a_8 \to \frac{1\,101\,935}{78\,983\,532}\}\}$

In[20]:= **TruthTable[Model312]**

Out[20]=

A	B	C	var	Pr
T	T	T	a_8	$\frac{1\,101\,935}{78\,983\,532}$
T	T	F	a_5	$\frac{4}{61}$
T	F	T	a_6	$\frac{1}{540}$
T	F	F	a_2	$\frac{1}{25}$
F	T	T	a_7	$\frac{5}{57}$
F	T	F	a_3	$\frac{27}{44}$
F	F	T	a_4	$\frac{3}{41}$
F	F	F	a_1	$\frac{92\,702\,054\,723}{890\,539\,323\,300}$

(b) show that — *if A and B are mutually exclusive* — *then* there can be no such probability distributions. In other words, show that the following *three* constraints *cannot* be jointly satisfied.

Pr(*C* | A) = Pr(*C* | B),
Pr(*C* | A) ≠ Pr(*C* | A v B),
Pr(A & B) = 0.

Again, easily solved with **PrSAT**.

In[21]:=
```
PrSAT[{
  Pr[C | A] == Pr[C | B],
  Pr[C | A] ≠ Pr[C | A ⋁ B],
  Pr[A ⋀ B] == 0
 }]
```

PrSAT :: srchfail : Search phase failed ; attempting FindInstance

Out[21]= { }

Accessing PrSAT's Underlying Algebraic Represenation

The Function **AlgebraicForm** allows you to work directly with the underlying algebraic representations of probability claims. For instance, we can look at the algebraic constraints corresponding to the set of constraints from the previous problem, as follows:

In[22]:=
```
AlgebraicForm[{
  Pr[C | A] == Pr[C | B],
  Pr[C | A] ≠ Pr[C | A ⋁ B],
  Pr[A ⋀ B] == 0
 }, {A, B, C}]
```

Out[22]= $$\left\{\frac{a_6 + a_8}{a_2 + a_5 + a_6 + a_8} == \frac{a_7 + a_8}{a_3 + a_5 + a_7 + a_8},\ \frac{a_6 + a_8}{a_2 + a_5 + a_6 + a_8} \neq \frac{a_6 + a_7 + a_8}{a_2 + a_3 + a_5 + a_6 + a_7 + a_8},\ a_5 + a_8 == 0\right\}$$

The function **PrKey** allows you to see a stochastic truth-table representation of the states involved for a given set of propositional letters.

In[23]:= **PrKey[{A, B, C}]**

Out[23]=

A	B	C	var	Pr
T	T	T	a_8	$\frac{1}{8}$
T	T	F	a_5	$\frac{1}{8}$
T	F	T	a_6	$\frac{1}{8}$
T	F	F	a_2	$\frac{1}{8}$
F	T	T	a_7	$\frac{1}{8}$
F	T	F	a_3	$\frac{1}{8}$
F	F	T	a_4	$\frac{1}{8}$
F	F	F	a_1	$\frac{1}{8}$